4. Gaussian derivatives

A difference which makes no difference is not a difference.
Mr. Spock (stardate 2822.3)

■ Init

```
<< MathVisionTools`;
Off[GaussianDerivative::"scalefail"];
Unprotect[gD]; Clear[gD]; gD[im_, nx_, ny_, σ_] :=
 Module[{t}, t = 1/2 σ^2; GaussianDerivative[{t, nx}, {t, ny}][im]]
Protect[gD];
SetOptions[RasterPlot, Frame -> False];
```

4.1 Shape and algebraic structure

When we take derivatives to *x* (*spatial derivatives*) of the Gaussian function repetitively, we see a pattern emerging of a polynomial of increasing order, multiplied with the original (normalized) Gaussian function again. Here we show a table of the derivatives from order 0 (i.e. no differentiation) to 3.

```
Unprotect[gauss]; gauss[x_, σ_] := 1/(σ Sqrt[2 π]) Exp[-x^2/(2 σ^2)];
Table[Factor[Evaluate[D[gauss[x, σ], {x, n}]]], {n, 0, 4}]
```

$$\left\{\frac{e^{-\frac{x^2}{2\sigma^2}}}{\sqrt{2\pi}\,\sigma},\ -\frac{e^{-\frac{x^2}{2\sigma^2}}\,x}{\sqrt{2\pi}\,\sigma^3},\ \frac{e^{-\frac{x^2}{2\sigma^2}}\,(x-\sigma)\,(x+\sigma)}{\sqrt{2\pi}\,\sigma^5},\ -\frac{e^{-\frac{x^2}{2\sigma^2}}\,x\,(x^2-3\sigma^2)}{\sqrt{2\pi}\,\sigma^7},\ \frac{e^{-\frac{x^2}{2\sigma^2}}\,(x^4-6x^2\sigma^2+3\sigma^4)}{\sqrt{2\pi}\,\sigma^9}\right\}$$

```
GraphicsGrid[Partition[Table[Plot[Evaluate[∂{x,n} gauss[x, 1]],
   {x, -5, 5}, PlotLabel → "Order=" <> ToString[n],
   DisplayFunction → Identity], {n, 0, 7}], 4], ImageSize → 500]
```



Figure 4.1 Plots of the 1D Gaussian derivative function for order 0 to 7.

The Gaussian function itself is a common element of all higher order derivatives. We extract the polynomials by dividing by the Gaussian function:

```
Table[Evaluate[D[gauss[x, σ], {x, n}] / gauss[x, σ]], {n, 0, 4}] // Simplify
```

$$\left\{1, -\frac{x}{\sigma^2}, \frac{x^2-\sigma^2}{\sigma^4}, -\frac{x^3-3x\sigma^2}{\sigma^6}, \frac{x^4-6x^2\sigma^2+3\sigma^4}{\sigma^8}\right\}$$

These polynomials are the Hermite polynomials, called after Charles Hermite, a brilliant French mathematician (see figure 4.2).

```
Show[Import["Charles Hermite.jpg"], ImageSize → 150]
```

Figure 4.2 Charles Hermite (1822-1901).

They emerge from the following definition: $\frac{\partial^n e^{-x^2}}{\partial x^n} = (-1)^n\, H_n(x)\, e^{-x^2}$. The function $H_n(x)$ is the Hermite polynomial, where n is called the order of the polynomial.

When we make the substitution $x \to x/(\sigma\sqrt{2})$, we get the following relation between the Gaussian function $G(x, \sigma)$ and its derivatives: $\frac{\partial^n G(x,\sigma)}{\partial x^n} = (-1)^n \frac{1}{(\sigma\sqrt{2})^n} H_n\left(\frac{x}{\sigma\sqrt{2}}\right) G(x, \sigma)$.

In *Mathematica* the function H_n is given by the function **HermiteH[n,x]**. Here are the Hermite functions from zeroth to fifth order:

```
Table[HermiteH[n, x], {n, 0, 7}] // TableForm
```

1
$2x$
$-2 + 4x^2$
$-12x + 8x^3$
$12 - 48x^2 + 16x^4$
$120x - 160x^3 + 32x^5$
$-120 + 720x^2 - 480x^4 + 64x^6$
$-1680x + 3360x^3 - 1344x^5 + 128x^7$

```
Clear[σ]; gd[x_, n_, σ_] := (-1/(σ Sqrt[2]))^n HermiteH[n, x/(σ Sqrt[2])] 1/(σ Sqrt[2 π]) Exp[-x^2/(2 σ^2)]
```

Check:

```
Simplify[gd[x, 4, σ], σ > 0]
```

$$\frac{e^{-\frac{x^2}{2\sigma^2}}\left(x^4-6x^2\sigma^2+3\sigma^4\right)}{\sqrt{2\pi}\ \sigma^9}$$

```
Simplify[D[1/(σ Sqrt[2 π]) Exp[-x^2/(2 σ^2)], {x, 4}], σ > 0]
```

$$\frac{e^{-\frac{x^2}{2\sigma^2}}\left(x^4-6x^2\sigma^2+3\sigma^4\right)}{\sqrt{2\pi}\ \sigma^9}$$

The amplitude of the Hermite polynomials explodes for large x, but the Gaussian envelop suppresses *any* polynomial function.

```
f[x_] := (1/Sqrt[2])^7 HermiteH[7, x/Sqrt[2]]; GraphicsRow[
 {Plot[f[x], {x, -5, 5}], p2 = Plot[f[x] e^(-x^2/2), {x, -5, 5}]}, ImageSize → 400]
```



Figure 4.3 Left: The 7th order Hermite polynomial. Right: idem, with a Gaussian envelop (weighting function). This is the 7th order Gaussian derivative kernel.

Due to the limiting extent of the Gaussian window function, the amplitude of the Gaussian derivative function can be negligible at the location of the larger zeros. We plot an example, showing the 20th order derivative and its Gaussian envelope function:

```
n = 20; σ = 1; Show[{Plot[gd[x, n, σ], {x, -5, 5}, Filling → Axis],
  Plot[gd[0, n, σ] gauss[x, σ] / gauss[0, σ], {x, -5, 5}]}, ImageSize → 200]
```



Figure 4.4 The 20th order Gaussian derivative's outer zero-crossings vahish in negligence. Note also that the amplitude of the Gaussian derivative function is not bounded by the Gaussian window. The Gabor kernels, as we will discuss later in section 4.XXX, are bounded by the Gaussian window.

4.2 Gaussian derivatives in the Fourier domain

The Fourier transform of the derivative of a function is $(-i\,\omega)$ times the Fourier transform of the function. For each differentiation, a new factor $(-i\,\omega)$ is added. So the Fourier transforms of the Gaussian function and its first and second order derivatives are:

```
σ =.; Simplify[FourierTransform[
  {gauss[x, σ], ∂x gauss[x, σ], ∂{x,2} gauss[x, σ]}, x, ω], σ > 0]
```

$$\left\{\frac{e^{-\frac{1}{2}\sigma^2\omega^2}}{\sqrt{2\pi}},\ -\frac{i\,e^{-\frac{1}{2}\sigma^2\omega^2}\,\omega}{\sqrt{2\pi}},\ -\frac{e^{-\frac{1}{2}\sigma^2\omega^2}\,\omega^2}{\sqrt{2\pi}}\right\}$$

In general: $\mathcal{F}\left\{\frac{\partial^n G(x,\sigma)}{\partial x^n}\right\} = (-\mathrm{i}\omega)^n\,\mathcal{F}\,\{G(x,\,\sigma)\}$.

The normalized powerspectra show that higher order of differentiation means a higher center frequency for the *bandpass* filter. The bandwidth remains virtually the same.

```
σ = 1; p1 = Plot[Table[Abs[ (-ⅈ ω)^n ⅇ^(-1/2 (σ^2 ω^2)) / ( (√(2π) ((-ⅈ √n)^n ⅇ^(-1/2 (σ^2 n)))) / √(2π) ) ], {n, 1, 12}],
  {ω, 0, 6}, PlotRange → All, AxesLabel → {"ω", ""}, ImageSize → 400]
```



Figure 4.5 Normalized power spectra for Gaussian derivative filters for order 1 to 12, lowest order is left-most graph, $\sigma = 1$. Gaussian derivative kernels act like bandpass filters.